# Maintainer: Felix Yan <felixonmars@archlinux.org>

# ALARM: Kevin Mihelich <kevin@archlinuxarm.org>
# - remove makedepend on python-pypandoc

pkgname=python-baron
pkgver=0.9
pkgrel=5
pkgdesc="Full Syntax Tree for python to make writing refactoring code a realist task"
url="https://github.com/PyCQA/baron"
license=('LGPL')
arch=('any')
depends=('python-rply')
makedepends=('python-setuptools')
checkdepends=('python-pytest-runner')
source=("$pkgname-$pkgver.tar.gz::https://github.com/PyCQA/baron/archive/$pkgver.tar.gz")
sha512sums=('ed31bad1442b8d868ab9872f057db31ad8730654a7e8d49119c75d5631c881b074baf5754b0b0b5023dad62eda1f40461e4445321695248cfa3530b608d48c8e')

build() {
  cd baron-$pkgver
  python setup.py build
}

check() {
  cd baron-$pkgver
  python setup.py pytest
}

package() {
  cd baron-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1
}
